
Quick start
    


  
    
  
  Start your first project


Nova provides a plugin to make working with the framework a lot easier. One can install it by either using
an automated installation or include it manually in the global rebar.config file.
Note :bangbang:
If you require help installing Erlang and/or rebar3 please check https://adoptingerlang.org/docs/development/setup/


  
    
  
  Automated installation


Via Curl
sh -c "$(curl -fsSL https://raw.githubusercontent.com/novaframework/rebar3_nova/master/install.sh)"

Via wget
sh -c "$(wget -O- https://raw.githubusercontent.com/novaframework/rebar3_nova/master/install.sh)"


  
    
  
  Manual installation


Open your rebar.config file that should reside in ~/.config/rebar3/rebar.config*. If the file does not exist you
can just create it. Locate the plugins section of the file and include the nova plugin. The result should look something like
the following:
{plugins,[{rebar3_nova,{git,"https://github.com/novaframework/rebar3_nova.git",
                            {branch,"master"}}}]}.

  
    
  
  Creating the skeleton


After the installation of the Nova-plugin is done use rebar3 to generate a new project.
rebar3 new nova my_first_nova

$ rebar3 new nova my_first_app
===> Writing my_first_nova/config/dev_sys.config.src
===> Writing my_first_nova/config/prod_sys.config.src
===> Writing my_first_nova/src/my_first_nova.app.src
===> Writing my_first_nova/src/my_first_nova_app.erl
===> Writing my_first_nova/src/my_first_nova_sup.erl
===> Writing my_first_nova/src/my_first_nova_router.erl
===> Writing my_first_nova/src/controllers/my_first_nova_main_controller.erl
===> Writing my_first_nova/rebar.config
===> Writing my_first_nova/config/vm.args.src
===> Writing my_first_nova/src/views/my_first_nova_main.dtl

Now the skeleton have been created and you should be able to start it. Go into the newly created directory and run the serve command.
Note :bangbang:
For the auto-compile/reload to work you need inotify to be installed.

$ cd my_first_app
$ rebar3 nova serve
===> Verifying dependencies...
===> Compiling my_first_app
Erlang/OTP 20 [erts-9.3] [source] [64-bit] [smp:8:8] [ds:8:8:10] [async-threads:0] [hipe] [kernel-poll:false]
...
a lot of progress-reports
...
===> Booted my_first_app
...

If you take a look at http://localhost:8080 you should be greeted by a Nova-page.


  

    
Configuration
    

There's a lot of parameters that can be configured in Nova. This document will try to explain them all.

  
    
  
  Cowboy configuration


Nova uses Cowboy as the webserver. Cowboy is a very flexible webserver and Nova tries to expose as much of this flexibility as possible. This means that you can configure Cowboy in a lot of different ways. The configuration is done in the nova-application under cowboy_configuration-key in your sys.config.
	Key	Description	Value	Default
	stream_handlers	Stream handlers are used to handle streaming requests. You can configure multiple stream handlers. Read more in the subsection Stream handlers	list()	[nova_stream_h, cowboy_compress_h, cowboy_stream_h]
	middleware_handlers	Middleware handlers are used to handle middleware requests. You can configure multiple middleware handlers. Read more in the subsection Middleware handlers	list()	[nova_router, nova_plugin_handler, nova_security_handler, nova_handler, nova_plugin_handler]
	options	Cowboy options. Read more in the subsection Cowboy options	map()	#{compress => true}
	ip	IP to bind to	tuple	{0, 0, 0, 0}
	port	Port to bind to	integer()	8080
	use_ssl	If SSL should be used	boolean()	false
	ssl_options	Transport options for SSL. Nova uses ranch_ssl module so read about available options on their page.	ranch_ssl:opts()	#{cert => "/path/to/cert.pem", key => "/path/to/key.pem"}
	ssl_port	Port to bind to when using SSL	integer()	8443
	ca_cert	Path to CA-cert Deprecated since 0.10.3 - Read with ssl_options	string()	undefined
	cert	Path to cert Deprecated since 0.10.3 - Replaced with ssl_options	string()	undefined


  
    
  
  Nova specific configurations


Following parameters should be defined under the nova-key in your sys.config.
	Key	Description	Value
	use_persistent_term	Use persistent_term module to store routing tree	boolean()
	use_stacktrace	If Nova should include stacktrace in error-pages	boolean()
	render_error_pages	If Nova should render error-pages for HTML-request	boolean()
	use_sessions	Turn off/on support for sessions	boolean()
	session_manager	Specifify a module to use as the session manager. Defaults to nova_session_ets	atom()
	use_strict_routing	If the routing module should work under the strict mode. Using strict mode will cause errors if non-deterministic paths are detected. This is a beta-function so use with caution.	boolean()
	bootstrap_application	Define which application to bootstrap with Nova. This should be the name of your application.	atom()
	cowboy_configuration	If you need some additional configuration done to Cowboy this is the place. Check nova_sup module to learn which keys that can be defined.	map()


  
    
  
  Application parameters


These parameters can be specified in your main application (Eg the one you've specified in the bootstrap-section).
	Key	Description	Value
	json_lib	JSON lib to use. Read more in the subsection Configure json lib	atom()

| watchers | Watchers are external programs that will run together with Nova. Watchers are defined as list of tuples where the tuples is in format {Command, ArgumentList} (Like [{my_app, "npm", ["run", "watch"], #{workdir => "priv/assets/js/my-app"}}]) | [{string(), string()}] | [{atom(), string(), map()}] | [{atom(), string(), list(), map()}] |

  
    
  
  Configure json_lib


One can configure which json library to use for encoding/decoding json structures. The module defined for this should expose two different functions:
encode(Structure) -> binary() | iolist()
decode(JsonString) -> {ok, Structure}

  
    
  
  Handling errors in Nova


Nova will by default render a error page if an error occurs. This page will be rendered using the nova_error-template. This template can be overridden by defining a template with the same name in your application.
By defauly Nova outputs a lot of information, including the stacktrace. This might not be a good approach in production. To turn off stacktraces in production you can add the following to your sys.config:
{nova, [{render_error_pages, false}]}
This will exclude stacktrace from the error page.
Note Nova is aware about which accept-headers the request is sent with and will respond with the correct content-type. If the request is sent with application/json Nova will respond with a JSON-structure instead of a HTML-page.


  

    
Routing
    

Each nova application have their own routes file. This file contains information about all the routes existing for this application.

  
    
  
  Basic components


A simple route file could look something like this:
-module(my_app_router).

-export([routes/1]).

routes(_Environment) ->
  [#{prefix => "/admin",
    security => false,
    routes => [
      {"/", fun my_controller:main/1, #{methods => [get]}}
    ]
  }].
This will create a path for /admin which, when a user enters will call my_controller:main/1. The _Environment variable that is consumed by the routes function will have the value that nova:get_environment/0 returns. The environment variable is an important thing cause it enables the devlopers to define different routes for different environments. To change the running environment edit the sys.config file to include an {environment, Env} tuple under the nova application where Env can be any erlang-term.

  
    
  
  The routing object


The routing object consists of three or four fields.

  
    
  
  HTTP(S) Routing


{Route :: list(), ControllerCallback :: function(), Options :: map()}
As you saw in the initial example in the Basic components section, we defined a route for the root path /.

  
    
  
  Websocket routing


{Route :: list(), Controller :: atom(), Options :: map()}
Important
One needs to define protocol => ws in the options-map in order to enable websocket communications.


  
    
  
  Static files


One can also define static files to be served by nova. This is done by adding a tuple to the route entries. The value of this tuple should be of size 2 or  3  where the first element is the url and the second element is the path to the file on the filesystem related from the apps priv directory. An additional third element can be added to the tuple to define options for this particular static file or directory.
Note! The notation [...] can be used as a wildcard (Zero or many occurences) in the url-section.
Valid options is;
	mimetype - which mimetype the file should be served as. If not defined nova will try to guess the mimetype based on the file extension.
	index_files - a list of filenames that can be used as an index. This is relevant if a directory is served.
	list_dir - Set to true if allowing the requester to list the content of a directory (if such is served)

Example:
{"/my/static/directory/[...]", "assets/a-local-dir", #{list_dir => true}},
{"/with-index/[...]", "assets/another-dir", #{index_files => ["index.html"]}}

  
    
  
  How to create routes


A route consists of three different components:
	Path - This is the actual path to the endpoint. Eg "/admin"
	Method - What method you want to support for this endpoint (get, post, update, delete). If you want to support all methods you can use the '_'-atom.
	ControllerCallback - What erlang callback should be called on when the path gets called. This is defined as a function reference, eg fun my_controller:main/1.


  
    
  
  Using prefix


You can group paths where you prefix them with a path. This is especially useful when having several different nova applications running. A very common example would be if you had an administration interface. Then the prefix would be "/admin". Another example is versioned APIs.
Prefix is defined at top level of a route-entry which can be seen in the Basic components section of this chapter.

  
    
  
  Secure routing


You can secure routes by providing a module and function to the security directive in the route entry. Here's a simple example of this:
#{prefix => "/admin",
  type => html,
  security => fun security_controller:do_security/1,
  routes => [
    {"/", fun my_controller:main/1, #{methods => [get]}}
  ]
}
This will cause nova to call security_controller:do_security/1 before calling the actual controller for all routes defined in the above route entry.
The security function should return a boolean (If the user can proceed or not).
An example of a security function:
do_security(Req) ->
    maps:get(host, Req) == "my_domain.com".
This will cause nova to return a 401 status code for all requests not coming from my_domain.com

  
    
  
  Using plugins local to a set of endpoints


It's possible to configure a small set of endpoints with a specific plugin. This is done by adding a plugins key to the route entry. The value of this key should be a list of plugins to use for this route entry.
#{prefix => "/admin",
  plugins => [
    {pre_request, nova_json_schemas, #{render_errors => true}}
  ],
  routes => [
    {"/", fun my_controller:main/1, #{methods => [get]}}
  ]
}
In the example above we have enabled the pre-request-plugin nova_json_schemas for all routes under the /admin prefix. This will cause all requests to be validated against the JSON schema defined in the nova_json_schemas plugin.
You can also include post-request-plugins in the same way.

  
    
  
  Adding routes programatically


You can also add routes programatically by calling nova_router:add_route/2. This is useful if you want to add routes dynamically. The spec for it is:
%% nova_router:add_route/2 specification
-spec add_route(App :: atom(), Routes :: map() | [map()]) -> ok.
First argument is the application you want to add the route to. The second argument is the route or a list of routes you want to add - it uses the same structure as in the regular routers.
nova_router:add_route(my_app, #{prefix => "/admin", routes => [{"/", fun my_controller:main/1, #{methods => [get]}}]}).
This will add the routes defined in the second argument to the my_app application.
Note: If a route already exists it will be overwritten.


  

    
Controllers
    

Controllers are central in how Nova works. They are in charge of handling all user-implemented logic for a request.
Controllers are located in the /src/controllers/ directory of your Nova application. A controller is basically a regular Erlang module that exposes functions you've provided in the routing file.
[image: Request life-cycle]

  
    
  
  Handlers


Handlers are modules that interprets the resulting output from a controller and sends it to the requester. Handlers are identified by the first atom in the return of a controller, eg {json, #{status => "ok"}} calls the json handler. You can read about the handlers below and what their functions are.

  
    
  
  JSON structures


Simple interface
Keyword: json
Spec: {json, Structure :: map()}
Example: {json, #{status => "ok"}}.
Converts the map (Second element in the resulting tuple) and sends it to the requester with HTTP-status code 200.
Advanced interface
Keyword: json
Spec: {json, StatusCode :: integer(), Headers :: map(), JSON :: map()}
Example: {json, 201, #{"x-correlation-id", "EX123"}, #{some_response => true}}
Same as the simple interface but with two additional elements for specifying HTTP-status code and additional headers.

  
    
  
  HTML templates (Using Erlydtl)


Nova renders templates using ErlyDTL (https://github.com/erlydtl/erlydtl which uses the django template language to express logic. To get a better overview for the functionality that comes with dtl check their documentation page
Simple interface
Keyword: view
Spec: {view, Variables :: map() | [{Key :: atom() | binary() | string(), Value :: any()}]}
Example: {view, #{my_var => "123"}}
Renders the corresponding view with the variables attached. If a controller is named my_simple_controller.erl the view is named my_simple_view.dtl.
Advanced interface
Keyword: view
Spec: {view, Variables :: map() | [{Key :: atom() | binary() | string(), Value :: any()}], Options :: map()}
Example: {view, #{my_var => "123"}, #{view => my_view_mod}}
Same as the simple interface but where you can define some options. Currently the only option for this interface is view which enables the user to specify a view other than the corresponding one based on controllers name. In the example above the dtl-file my_view_mod.dtl would be rendered.

  
    
  
  HTTP-status codes


Simple interface
Keyword: status
Spec: {status, StatusCode :: integer()}
Example: {status, 200}
Returns a HTTP-code to the requester with an empty body.
Medium interface
Keyword: status
Spec: {status, StatusCode :: integer(), ExtraHeaders :: map()}
Example: {status, 200, #{"content-type" => "application/json"}}
Same as the simple interface but with an additional field for specifying additional headers.
Advanced interface
Keyword: status
Spec: {status, Status :: integer(), ExtraHeaders :: map(), Body :: binary()}
Example: {status, 200, #{"content-type" => "text/plain"}, "A plain text"}
Same as the medium interface but with an additional field for specifying a body.

  
    
  
  File transfers (Using cowboys sendfile functionality)


Keyword: sendfile
Spec: {sendfile, StatusCode :: integer(), Headers :: map(), {Offset :: integer(), Length :: integer(), Path :: list()}, Mime :: binary()}
Example: {sendfile, 200, #{}, {0, 12345, "path/to/logo.png"}, "image/png"}
Sends a file using sendfile. This uses cowboys sendfile functionality and more information about it can be found in the cowboy manual on sendfile

  
    
  
  Redirecting user


Keyword: redirect
Spec: {redirect, Route :: list() | binary()}
Example: {redirect, "/my/other/path}
Sends a temporary redirect (HTTP status code 302) for the specified path to requester.

  
    
  
  Fallback controllers


Phoenix have a really useful feature which they call action_fallback. We thought it would be a good addition to Nova to include something similar and therefore the fallback_controller was introduced. If a controller returns an invalid/unhandled result the fallback controller gets invoked and can take action on the payload. It's good for separating error-handling from the controllers. A fallback controller is set by setting the fallback_controller attribute with the module name of the fallback controller.
The following example shows how a controller defines a fallback
-module(my_main_controller).
-export([
    error_example/1
]).
-fallback_controller(my_fallback_controller).

error_example(_Req) ->
    %% Since {error, ...} is not a valid handler the fallback-controller will be invoked
    {error, example_error}.
A fallback controller exposes one function resolve/2 which returns a handler (like for regular controllers) in order to return the response to client. If we take the previous example and try and build a fallback controller for it:
-module(my_fallback_controller).
-export([
    resolve/2
]).

resolve(Req, {error, example_error}) ->
  {status, 400}.

  
    
  
  Plugins


Plugins are part of the Nova core and can be executed both before and after the execution of a controller. They can both terminate a request early (like the request plugin does) or transform data into another structure (json plugin). Plugins can be defined in two different places; Global plugins are defined in the sys.config file and will be executed for every incoming request. If a plugin should only be executed for a limited set of endpoints it can be defined in the router file for that specific application (we call there local plugins).

  
    
  
  Global plugins


Global plugins are defined in the sys.config file and can have two different states: pre and post controller. They are exectued before or after a controller. Global plugins lives under the nova application, plugins key. An example of a sys.config file:
{nova, [
  {plugins, [
    {pre_request, [
      {nova_request_plugin, #{decode_json_body => true}}
    ]},
    {post_request, [
    ]}
  ]}
]}.
In order to find the valid options one can call the plugin_info/0 function of each plugin. Currently there are four plugins shipped with Nova.
NOTE
If you are creating an application that can be included in another Nova application all of your plugins should be defined in the router-file (Local plugins) in order to avoid being overwritten.


  
    
  
  Local plugins


Local plugins works almost as the global ones but for a limited set of paths. Local plugins are therefore declared in the router-file for each group of endpoints. They are declared in the same way as the global ones.
Example:
routes(_Env) ->
  [#{
    prefix => "/api/json",
    security => false,
    plugins => [
      {pre_request, [
        {nova_request_plugin, #{decode_json_body => true}}
      ]}
    ],
    routes => [
      {"/my_json_route", {my_app_json_controller, json}, #{methods => [get, post]}}
    ]
  }].
It's recommended to use this method of specifying routes if you plan to use it as a component in another Nova application.

  
    
  
  Websockets


Coming soon

  
    
  
  Callbacks


Coming soon


  

    
Views
    

The views in Nova are powered by erlydtl. They use the Django template language and compile down to Erlang.
Please read the erlydtl wiki for more information regarding the templates.


  

    
Handlers
    


  
    
  
  Handlers


Handlers are a nifty thing that is called on when the controller returns. For example if the controller returns
{json, #{hello => world}} we would like Nova to create a proper JSON response to the requester. That means setting correct
headers, encode the payload and send it out. This is what handlers are for. They are (often short) functions that transforms something like
{json, Payload} to proper output.


  

    
Plugins
    

Plugins are a bit like the handlers except they are run on request. There's currently two different type of plugins; pre_request and post_request.
These can be used to create access logs, insert CORS headers or similar.
Plugins are used to handle things before and/or after a request. They are applied on all requests of a specified protocol.
This is an example:
-module(correlation_id).
-behaviour(nova_plugin).
-export([
    pre_request/2,
    post_request/2,
    plugin_info/0
  ]).

pre_request(Req, NovaState) ->
    UUID = uuid:uuid_to_string(uuid:get_v4()),
    {ok, cowboy_req:set_resp_header(<<"x-correlation-id">>, UUID, Req), NovaState}.

post_request(Req, NovaState) ->
    {ok, Req, NovaState}.

plugin_info() ->
    {<<"Correlation plugin">>, <<"1.0.0">>, <<"Niclas Axelsson <niclas@burbas.se>">>,
     <<"Example plugin for nova">>}.
This plugin injects a UUID into the headers.
Adding a plugin
Example:
A good example of a very useful plugin is the nova_request_plugin. When we are developing a HTTP web api using json as the data format, we need the framework to
decode our message so that we can process it. To do that we need to add decode_json_body => true into the options field in our sys.config.
sys.config
     {nova, [
         {environment, dev},
         {cowboy_configuration, #{
                                  port => 8080
                                 }},
         {dev_mode, true},
         {bootstrap_application, chatapp},
         {plugins, [
                    {pre_request, nova_request_plugin, #{parse_bindings => true,
                                                         decode_json_body => true}}
                   ]}
        ]}
 We have added our plugin in the plugins section. As we can see this is a pre_request plugin since it processes and decodes the message to json format
 before we can actually use it in our nova application endpoints.
Usage:
controller
-module(test_controller).
-export([increment/1]).

increment(#{<<"json">> := #{<<"id">> := Id, <<"value">> := Value}})->
    {json,200,#{},#{<<"id">> => Id , <<"received">> => Value, <<"increment">> => Value+1}}.


  
    
  
  Nova plugins


Nova has a couple of plugins for some general purposes.
	Plugin	Description	Code
	nova_correlation_plugin	This plugin will add a correlation id to header response but also add #{correlation_id => CorrelationID} to the request obj that is passed to the controller.	nova_correlation_plugin
	nova_cors_plugin	This plugin will handle cors and add the cors headers into the request.	nova_cors_plugin
	nova_request_plugin	This plugin will handle incomming data like qs, form urlencoded and json	nova_request_plugin


  
    
  
  Nova correlation


This plugin will generate a uuid v4 and set it as a response header as X-Correlation-ID if nothing is configuered.
{pre_request; nova_correlation_plugin, #{request_correlation_header => CorrelationHeader,
                                         logger_metadata_key => LoggerMetaDataKey}}
	Option	Description
	request_correlation_header	This is if you want a different correlation header than the standard X-Correlation-ID
	logger_metadata_key	This is if you want to have a different metadata key then the standard correaltion_id


  
    
  
  Nova cors


This plugins will make it so that if we get method OPTIONS it will just return back the CORS headers. In this case you don't need a controller to handle it and the plugin stops after this.
For other methods it will add the CORS headers to the request.
{pre_request; nova_cors_plugin, #{allow_origins => <<"*">>}}
	Option	Description
	allow_origins	Specifies which origins to insert into Access-Control-Allow-Origin


  
    
  
  Nova request


This plugins handle incoming data and can transform them to erlang maps depending on what the options are.
{pre_request; nova_correlation_plugin, #{decode_json_body => true,
                                         read_urlencoded_body => true,
                                         parse_qs => true|list}}
	Option	Description	Req
	decode_json_body	If header is application/json it will decode the body.	Req#{json => Map}
	read_urlencoded_body	If header is application/x-www-form-urlencoded it will decode it.	Req#{params => Map}
	parse_qs	If the path have qs in it we will get them.	Req#{parsed_qs => Map or List}



  

    
Nova Pubsub
    

With version 0.9.4 we introduced the concept of a pubsub-like mechanism in Nova so that users can build distributed services. Internally it relies on Erlangs pg2-module which enables distributed named process groups. The concept is really simple and there's not much depth in the current implementation - it should only be used as a simple message bus. If you need more advanced features please take a look at RabbitMQ or MQTT.

  
    
  
  Basic concepts


The idea is that a process subscribes to a topic. If there's any messages sent to this topic all the subscribed processes receivs that message.
[image: Processes]
This picture shows how two processes have subscribed to a topic (Or channel) "Messages". When another process then sends "Hello"* on this topic the other two will receive it.

  
    
  
  Example


-module(test_module).
-export([player1/0,
         player2/0,
         start_game/0]).

player1() ->
  spawn(fun() ->
    nova_pubsub:join(game_of_pong),
    game_loop(1, "pong", "ping")
  end).

player2() ->
  spawn(fun() ->
    nova_pubsub:join(game_of_pong),
    game_loop(2, "ping", "pong")
  end).

game_loop(Player, ExpectedMessage, Smash) ->
  receive
    ExpectedMessage ->
      io:format("Player ~d received ~s and returning ~s~n", [Player, ExpectedMessage, Smash]),
      nova_pubsub:broadcast(game_of_pong, "match1", Smash),
      game_loop(Player, ExpectedMessage, Smash);
  _ ->
      game_loop(Player, ExpectedMessage, Smash)
  end.
Here we are subscribing to a topic "game_of_pong" which we later use to send either ping or pong to, depending on which player that are serving.
This can ofcourse be extended to do so much more, but this is a taste of what one can use nova_pubsub to.

  
    
  
  Configuration


It does not require any specific parameters set in order to work. Just remember to enable distributed erlang.


  

    
Building releases
    

Nova utilizes relx for building releases and has a particular profile defined for creating such. In a standard Nova-project there will be two profiles defined; devand prod. The dev-profile is used for local development and the prod-profile is used for building releases.
So in order to build a release for production, you would run the following command:
$ rebar3 as prod tar

This builds a release and archives it into a tar-file. As default erts is included in the release, but this can be configured through the options in rebar.config. To read more about release handling and configuration options we recommend reading relx-manual.
If you are inexperienced with building releases we recommend reading the relx-manual.


  

    
Books and links
    


  
    
  
  Online books


	Learn you some Erlang


  
    
  
  Books


	Learn you some Erlang
	Erlang OTP in action


  
    
  
  Blogs & Articles


	Nova a web framework for Erlang - Daniel Widgren, Niclas Axelsson | Code BEAM V 2020

	Nova in Hello Erlang-podcast
	Getting around with NOVA | Daniel Widgren, Niclas Axelsson | Code BEAM America 2022

	Fireside chat on Nova and Nitrogen | Daniel Widgren, Niclas Axelsson & Jesse Gumm | Code BEAM V EU21




  

    
Rebar3 Nova
    

If you have used the installation script, documented in the main README.md-file you should have a working installation of rebar3 nova. This gives you some handy commands to work with Nova;
rebar3 new nova <project-name> - Creates a new Nova project in the directory <project-name>
rebar3 nova serve - Starts a local webserver on port given in sys.config, serving the current project. This means that if something is changed in the project, the server will automatically reload the changed files.
rebar3 nova routes - Lists all routes in the current project (Includes included Nova applications aswell if they are properly configured)
We are hoping to increase the amount of commands that is available from the rebar3 nova command in the future.

  Plugin Configuration


To not break backwards compatibility in a minor release, some behavior is behind configuration items.

  
    
  
  Request Correlation Header Name


Set request_correlation_header in the plugin config to read the correlation ID from the request headers.
Notice: Cowboy request headers are always in lowercase.

  
    
  
  Default Correlation ID Generation


If the header name is not defined or the request lacks a correlation ID header, then the plugin generates a v4 UUID automatically.

  
    
  
  Logger Metadata Key Override


Use logger_metadata_key to customize the correlation ID key in OTP logger process metadata. By default it is set to <<"correlation-id">>.

  
    
  
  Correlation ID in Request Object


The plugin defines a field called correlation_id in the request object for controller use if it makes further requests that it want to pass on the correlation id to.

  
    
  
  Example configuration


       {plugins, [
           {pre_request, nova_correlation_plugin, #{
               request_correlation_header => <<"x-correlation-id">>,
               logger_metadata_key => correlation_id
           }}
       ]}

This module is responsible for all the different return types a controller have. Nova is constructed in such way that it's really easy to extend it by using handlers. A handler is basically a module consisting of a function of arity 4. We will show an example of this.
If you implement the following module:
-module(my_handler). -export([init/0, handle_console]).
init() -> nova_handlers:register_handler(console, {my_handler, handle_console}).
handle_console({console, Format, Args}, {Module, Function}, State) -> io:format("~n=====================~n", []). io:format("~p:~p was called.~n", []), io:format("State: ~p~n", [State]), io:format(Format, Args), io:format("~n=====================~n", []), {ok, 200, #{}, EmptyBinary}.
The init/0 should be invoked from your applications supervisor and will register the module my_handler as handler of the return type {console, Format, Args}. This means that you can return this tuple in a controller which invokes my_handler:handle_console/4.
A handler can return two different types
{ok, StatusCode, Headers, Body} - This will return a proper reply to the requester.
{error, Reason} - This will render a 500 page to the user.

Plugins can be run at two different times; either in the beginning or at the end of a request. They can modify both the actual request or the nova-state. A plugin is implemented with the nova_plugin behaviour and needs to implement three different functions: pre_request/2, post_request/2 and plugin_info/0.
A plugin can return either {ok, NewState}, {break, NewState}, {stop, NewState}, {error, Reason}.
{ok, NewState} will continue the normal execution with the NewState.
{break, NewState} breaks the execution of the current plugin-chain. This means that if a pre_request-plugin returns the break-statement the rest of the plugins in that chain will be skipped.
{stop, NewState} stops the execution. The plugin is responsible in this case for returning a proper response to the client.
{error, Reason} will stop the execution and call the nova_error plugin, resulting in a 500 response back to the user. If debug mode is enabled the reason will be returned in the 500-response.

Pubsub system for Nova. It uses the pg/pg2 module.
Pubsub subsystem is started with Nova and does not need any additional configuration. It uses the pg/pg2 module depending on the version of OTP. It provides a simple way of distributing messages to a large set of receivers and exposes a simple set of functions for doing that.
A simple example of how to use pubsub in a ping/pong inspired game engine:
-module(test_module). -export([player1/0, player2/0, start_game/0]).
player1() -> spawn(fun() -> nova_pubsub:join(game_of_pong), game_loop(1, "pong", "ping").
player2() -> spawn(fun() -> nova_pubsub:join(game_of_pong), game_loop(2, "ping", "pong").
game_loop(Player, ExpectedMessage, Smash) -> receive ExpectedMessage -> io:format("Player ~d received ~s and returning ~s~n", [Player, ExpectedMessage, Smash]), nova_pubsub:broadcast(game_of_pong, "match1", Smash), game_loop(Player, ExpectedMessage, Smash); _ -> game_loop(Player, ExpectedMessage, Smash) end.

